
Using Maple's Groebner Basis Package.
with(Groebner);

The commands that we will mainly use are 
Basis - for computing a Groebner basis
NormalForm - for computing the remainder of a polynomial divided by a (Groebner) basis
Lets execute Buchberger's algorithm on the ideal I = < , > below.
.
f1 := x*y-y^2;
f2 := x^3-z^2;
G0 := [f1,f2];

We'll use the SPolynomial and LeadingMonomial commands.
And we'll use lexicographical order with  
LeadingMonomial(f1,plex(x,y,z));
LeadingMonomial(f2,plex(x,y,z));

f3 := SPolynomial(f1,f2,plex(x,y,z));

The NormalForm computes the remainder of S(f1,f2) divided by G.

f3 := NormalForm(f3,G0,plex(x,y,z));

The remainder is not 0 so we add f3 to the basis.
G1 := [f1,f2,f3];

map(LeadingMonomial,G1,plex(x,y,z));

Now S(f2,f3) reduces to 0 by Proposition 4 of 2.9 so we only need to consider
f4 := SPolynomial(f1,f3,plex(x,y,z));

f4 := NormalForm(f4,G1,plex(x,y,z));

So G1 is a Groebner basis for I = <f1,f2>.  It happens to be also reduced.  Let's check with 
Maple.
G := Basis([f1,f2],plex(x,y,z));

G1;

This polynomial is in the ideal I.  Let's test this
f := expand(x*f1+y*f2);

NormalForm(f,G,plex(x,y,z));
0

NormalForm(f+x+1,G,plex(x,y,z));

Monomial Orderings in Maple

Cox, Little O'Shea 
text

Maple

Lexicographical order lex with p l e x ( x , y , z )

Graded lexicographical 
order

grlex with g r l e x ( x , y ,
z )

Graded reverse 
lexicographical order

grevlex with 
 

t d e g ( x , y , z )

f := 3*x^3 + 4*x*y*z^2 + 5*y^3*z;

Notice that the LeadingTerm command in Maple returns a pair (leading coefficient, 
LeadingMonomial(f,plex(x,y,z));
LeadingCoefficient(f,plex(x,y,z));
LeadingTerm(f,plex(x,y,z));

LeadingMonomial(f,grlex(x,y,z));
LeadingCoefficient(f,grlex(x,y,z));

LeadingMonomial(f,tdeg(x,y,z));
LeadingCoefficient(f,tdeg(x,y,z));

5

Graph Coloring
Let's try to color the graph C3, a cycle on three vertices with k=2 colors (it's not 2-
colorable).

    

k := 2;
S := [x1^k-1,x2^2-1,x3^2-1];

The colors are the roots of  which are 1 and -1.  These equations say each vertex can 
be colored with either color.  But vertex 1 may not be the same color as vertex 2.  So we 
should add the equation x1 + x2 = 0.  Same for the other two edges.  We have
S := [op(S), x1+x2, x1+x3, x2+x3 ];

Basis(S,grlex(x1,x2,x3));

This means G is not 2-colorable.  But it is 3-colorable.
k := 3;
S := [x1^k-1,x2^k-1,x3^k-1,
      normal((x1^k-x2^k)/(x1-x2)),
      normal((x1^k-x3^k)/(x1-x3)),
      normal((x2^k-x3^k)/(x2-x3))];

Basis(S,grlex(x1,x2,x3));

This means x3 can be any color, x2 must be different from x3, and x1, x2, x3 must have all 
different colors.

Solving Polynomial Equations using lex Groebner bases.
Lets work with one of the ideals we saw in class, namely 

  I  = <  >
F := [x^2+y+z-1,x+y^2+z-1,x+y+z^2-1];

G := Basis(F,grlex(x,y,z));

H := Basis(F,plex(x,y,z));

We can see that the first polynomial has repeated roots.
factor(H[1]);

The Solve command in the Groebner basis package drops multiple solutions.
Solve(H,[x,y,z]);

This has split up the solutions.  We can solve them explicitly using solve
_EnvExplicit := true;
V := {solve(H,{x,y,z})};

nops(V);
5

There are 5 distinct solutions in the variety V.

The PolynomialIdeals package.
The PolynomialIdeals package has additional operations for ideals.
It also allows me to use < > brackets for ideals.
with(PolynomialIdeals);

interface(imaginaryunit = iii);
i i i

The above is to let me use the capital letter I for an ideal.
F;

I := <F>;

The radical operation gets rid of all repeated solutions.
J := Radical(I);

Compute a Groebner basis for J
Basis(J,plex(x,y,z));

The PrimeDecompostion operation splits the ideal J into prime components, each of which 
corresponds to an irreducible variety.
P := [PrimeDecomposition(J)];

P := map(Simplify,P);

From which we can see 1, 1, 2, 1 solutions.  From which we can understand the output of 
the Solve command above.


